

From well-formed substring tables to active charts	Overview	Representing active chart items																		
<p>Detmar Meurers: Intro to Computational Linguistics I OSU, LING 684.01</p> <p>Dotted rule examples</p> <ul style="list-style-type: none"> A dotted rule represents a state in processing a rule. Each dotted rule is a hypothesis: <table border="1" data-bbox="143 812 587 922"> <tr> <td style="padding: 5px;"> $vp \rightarrow \bullet v\text{-ditr} np pp\text{-to}$ $vp \rightarrow v\text{-ditr} \bullet np pp\text{-to}$ $vp \rightarrow v\text{-ditr} np \bullet pp\text{-to}$ $vp \rightarrow v\text{-ditr} np pp\text{-to} \bullet$ </td><td style="padding: 5px; vertical-align: top;"> We found a <i>vp</i> if we still find a <i>v-ditr</i>, a <i>np</i>, and a <i>pp-to</i> a <i>np</i> and a <i>pp-to</i> a <i>pp-to</i> nothing </td></tr> </table> <p>The first three are examples of active items (or active edges) The last one is a passive item/edge.</p>	$vp \rightarrow \bullet v\text{-ditr} np pp\text{-to}$ $vp \rightarrow v\text{-ditr} \bullet np pp\text{-to}$ $vp \rightarrow v\text{-ditr} np \bullet pp\text{-to}$ $vp \rightarrow v\text{-ditr} np pp\text{-to} \bullet$	We found a <i>vp</i> if we still find a <i>v-ditr</i> , a <i>np</i> , and a <i>pp-to</i> a <i>np</i> and a <i>pp-to</i> a <i>pp-to</i> nothing	<ul style="list-style-type: none"> CKY algorithm: <ul style="list-style-type: none"> explores all analyses in parallel bottom-up stores complete subresults desiderata: <ul style="list-style-type: none"> add top-down guidance (to only use rules derivable from start-symbol), but avoid left-recursion problem of top-down parsing store partial analyses (useful for rules right-hand sides longer than 2) Idea: also store partial results, so that the chart contains <ul style="list-style-type: none"> passive items: complete results active items: partial results 	<p>2</p> <ul style="list-style-type: none"> well-formed substring entry: $\text{chart}(i, j, A)$: from i to j there is a constituent of category A More elaborate data structure needed to store partial results: <ul style="list-style-type: none"> rule considered + how far processing has succeeded dotted rule: $i[A \rightarrow \alpha \bullet_j \beta]$ with $A \in N$ and $\alpha, \beta \in (\Sigma \cup N)^*$ active chart entry: $\text{chart}(i, j, \text{state}(A, \beta))$ Note that α is not represented. <p>3</p>																
$vp \rightarrow \bullet v\text{-ditr} np pp\text{-to}$ $vp \rightarrow v\text{-ditr} \bullet np pp\text{-to}$ $vp \rightarrow v\text{-ditr} np \bullet pp\text{-to}$ $vp \rightarrow v\text{-ditr} np pp\text{-to} \bullet$	We found a <i>vp</i> if we still find a <i>v-ditr</i> , a <i>np</i> , and a <i>pp-to</i> a <i>np</i> and a <i>pp-to</i> a <i>pp-to</i> nothing																			
<p>Eliminating scanning</p> <p>Scanning: for each $i[A \rightarrow \alpha \bullet_{j-1} w_j \beta]$ in chart add $i[A \rightarrow \alpha w_j \bullet_j \beta]$ to chart</p> <p>Completion: for each $i[A \rightarrow \alpha \bullet_k B \beta]$ and $k[B \rightarrow \gamma \bullet_j]$ in chart add $i[A \rightarrow \alpha B \bullet_j \beta]$ to chart</p> <p>Observation: Scanning = completion + words as passive edges. One can thus simplify scanning to adding a passive edge for each word: for each w_j in $w_1 \dots w_n$ add $j-1[w_j \rightarrow \bullet_j]$ to chart</p>	<p>The three actions in Earley's algorithm</p> <p>In $i[A \rightarrow \alpha \bullet_j B \beta]$ we call B the <i>active constituent</i>.</p> <ul style="list-style-type: none"> Prediction: Search all rules realizing the active constituent. Scanning: Scan over each word in the input string. Completion: Combine an active edge with each passive edge covering its active constituent. 	<p>A closer look at the three actions</p> <p>Prediction: for each $i[A \rightarrow \alpha \bullet_j B \beta]$ in chart for each $B \rightarrow \gamma$ in rules add $j[B \rightarrow \bullet_j \gamma]$ to chart</p> <p>Scanning: let $w_1 \dots w_j \dots w_n$ be the input string for each $j[A \rightarrow \alpha \bullet_{j-1} w_j \beta]$ in chart add $i[A \rightarrow \alpha w_j \bullet_j \beta]$ to chart</p> <p>Completion (fundamental rule of chart parsing): for each $i[A \rightarrow \alpha \bullet_k B \beta]$ and $k[B \rightarrow \gamma \bullet_j]$ in chart add $i[A \rightarrow \alpha B \bullet_j \beta]$ to chart</p>																		
<p>7</p>	<p>Earley's algorithm without scanning</p> <p>General setup: apply prediction and completion to every item added to chart</p> <p>Start: add $0[\text{start} \rightarrow \bullet_0 s]$ to chart for each w_j in $w_1 \dots w_n$ add $j-1[w_j \rightarrow \bullet_j]$ to chart</p> <p>Success state: $0[\text{start} \rightarrow s \bullet_n]$</p>	<p>A tiny example grammar</p> <p>Lexicon:</p> <table> <tr><td>vp</td><td>\rightarrow</td><td>left</td></tr> <tr><td>det</td><td>\rightarrow</td><td>the</td></tr> <tr><td>n</td><td>\rightarrow</td><td>boy</td></tr> <tr><td>n</td><td>\rightarrow</td><td>girl</td></tr> </table> <p>Syntactic rules:</p> <table> <tr><td>s</td><td>\rightarrow</td><td>$np \ vp$</td></tr> <tr><td>np</td><td>\rightarrow</td><td>$det \ n$</td></tr> </table> <p>8</p> <p>9</p>	vp	\rightarrow	left	det	\rightarrow	the	n	\rightarrow	boy	n	\rightarrow	girl	s	\rightarrow	$np \ vp$	np	\rightarrow	$det \ n$
vp	\rightarrow	left																		
det	\rightarrow	the																		
n	\rightarrow	boy																		
n	\rightarrow	girl																		
s	\rightarrow	$np \ vp$																		
np	\rightarrow	$det \ n$																		

An example run

```

start          1. 0[start → •0 s]
predict from 1 2. 0[s → •0 np vp]
predict from 2 3. 0[np → •0 det n]
predict from 3 4. 0[det → •0 the]
scan "the"
complete 4 with 5 5. 0[the → •1]
complete 3 with 6 6. 0[det → det •1]
predict from 7 7. 0[np → det •1 n]
predict from 7 8. 1[n → •1 boy]
9. 1[n → •1 girl]
scan "boy"
10. 1[boy → •2]
complete 8 with 10 11. 1[n → boy •2]
complete 7 with 11 12. 0[np → det n •2]
complete 2 with 12 13. 0[s → np •2 vp]
predict from 13 14. 2[vp → •2 left]
scan "left"
15. 2[left → •3]
complete 14 with 15 16. 2[vp → left •3]
complete 13 with 16 17. 0[s → np vp •3]
complete 1 with 17 18. 0[start → s•3]

```

10

The Earley algorithm in Prolog (parser/earley/earley.pl)

```

:- dynamic chart/3.           % chart(From,To,state(Lhs,Rest_Rhs))
:- op(1200,xfx,'-->').      % operator for grammar rules

% recognize(+WordList,+Startsymbol): Earley recognizer toplevel

recognize(String,Startsymbol) :-
    retractall(chart(_,_,_)),
    enter_edge(0,0,state('S',[Startsymbol])),
    scan(String,0,N),
    chart(0,N,state('S',[])).

```

11

```

% enter_edge(+FromIndex,+ToIndex,+Contents)

% a) only add if it does not yet exist:
enter_edge(I,J,State) :-
    chart(I,J,State),
    !.

% b) add to chart and make try prediction/completion
enter_edge(I,J,State) :-
    assertz(chart(I,J,State)),
    predict(I,J,State),
    complete(I,J,State).

```

12

```

predict(_,J,State) :-
    State = state(_,[_|_]),      % active edge
    (B --> Gamma),
    enter_edge(J,J,state(B,Gamma)),
    fail
; true.

% ----

complete(K,J,State) :-
    State = state(B,[]),        % passive edge
    chart(I,K,state(A,[B|Beta])),
    enter_edge(I,J,state(A,Beta)),
    fail
; true.

```

13

```

scan([],N,N).
scan([W|Ws],JminOne,N) :-
    J is JminOne+1,
    enter_edge(JminOne,J,state(W,[])),
    scan(Ws,J,N).

```

14

The tiny example grammar (parser/earley/earley_grammar.pl)

```

% lexicon:
vp --> [left].
det --> [the].
n --> [boy].
n --> [girl].

```

```

% syntactic rules:
s --> [np, vp].
np --> [det, n].

```

15

The example run in Prolog

(parser/parser/earley/earley_trace.pl, grammar: parser/earley/earley_grammar.pl)

```

I ?- recognize([the,boy,left]).  

START:   1: 0-state(S,[s])-----0  

PRED s in 1: 2: 0-state(s,[np, vp])----0  

PRED np in 2: 3: 0-state(np,[det,n])---0  

PRED det in 3: 4: 0-state(det,[the])---0  

SCAN 1 ([the]): 5: 0-state(the,[])-----1  

COMP 4 + 5: 6: 0-state(det,[])-----1  

COMP 3 + 6: 7: 0-state(np,[n])-----1  

PRED n in 7: 8: 1-state(n,[boy])-----1  

PRED n in 7: 9: 1-state(n,[girl])-----1  

SCAN 2 ([boy]): 10: 1-state(boy,[])-----2  

COMP 8 + 10: 11: 1-state(n,[])-----2  

COMP 7 + 11: 12: 0-state(np,[])-----2  

COMP 2 + 12: 13: 0-state(s,[vp])-----2  

PRED vp in 13: 14: 2-state(vp,[left])----2  

SCAN 3 ([left]): 15: 2-state(left,[])-----3  

COMP 14 + 15: 16: 2-state(vp,[])-----3  

COMP 13 + 16: 17: 0-state(s,[])-----3  

COMP 1 + 17: 18: 0-state(s,[])-----3  

SUCCESS: 18

```

16

Improving the efficiency of lexical access

- In the setup just described
 - words are stored as passive items so that
 - prediction is used for preterminal categories. The set of predicted words for a preterminal can be huge.
- If each word in the grammar is introduced by a preterminal rule *cat* → *word* one can add a **passive item for each preterminal category** which can dominate the word instead of for the word itself.
- What needs to be done:
 - syntactically distinguish syntactic rules (--->/2) from rules with preterminals on the left-hand side, i.e. lexical entries (lex/2).
 - modify scanning to take lexical entries into account

17

Code change for preterminals as passive edges (parser/earley/preterminals/earley.pl)

```

scan([W|Ws],JminOne,N) :-
    J is JminOne+1,
    enter_edge(JminOne,J,state(W,[])),
    scan(Ws,J,N).

```

is changed to

```

scan([W|Ws],JminOne,N) :-
    J is JminOne+1,
    ( lex(Cat,W),
      enter_edge(JminOne,J,state(Cat,[])),
      fail
    ; scan(Ws,J,N)).

```

18

The tiny example grammar in the modified format
(parser/earley/preterminals/grammar1.pl)

```
% lexicon:  
lex(vp, left).  
lex(det, the).  
lex(n, boy).  
lex(n, girl).  
  
% syntactic rules:  
s ---> [np, vp].  
np ---> [det, n].
```

19

The improved example run
(parser parser/earley/preterminals/earley_trace.pl, grammar: parser/earley/preterminals/grammar1.pl)

```
| ?- recognize([the,boy,left],S).  
START: 1: 0--state(S,[s])-----0  
PRED s in 1: 2: 0--state(s,[np, vp])---0  
PRED np in 2: 3: 0--state(np,[det, n])--0  
SCAN 1 (the): 4: 0--state(det,[])-----1  
COMP 3 + 4: 5: 0--state(np,[n])-----1  
SCAN 2 (boy): 6: 1--state(n,[])-----2  
COMP 5 + 6: 7: 0--state(np,[])-----2  
COMP 2 + 7: 8: 0--state(s,[vp])-----2  
SCAN 3 (left): 9: 2--state(vp,[])-----3  
COMP 8 + 9: 10: 0--state(s,[])-----3  
COMP 1 + 10: 11: 0--state(S,[])-----3  
SUCCESS: 11
```

20

Towards more flexible control

The algorithms, we saw

- use the Prolog database to store the chart and
- Prolog backtracking on edges in chart instead of an explicit agenda.

Alternatively, one can

- explicitly introduce an agenda
- to store and work off edges in any order one likes.

21

Earley-recognizer with explicit agenda and chart
(parser/earley/agenda/earley.pl)

```
:- op(1200, xfx, '--->'). % Operator for grammar rules  
  
% Data structures: chart(From,To,Category)  
% -----  
% recognize(+WordList)  
% top-level predicate for Earley recognizer  
  
recognize(String,Startsymbol) :-  
    StartAgenda=[chart(0,0,state('S',[Startsymbol]))],  
    process_agenda(StartAgenda,[],Chart0),  
    scan(String,0,N,Chart0,Chart),  
    element(chart(0,N,state('S',[])),Chart).
```

22

```
% process_agenda(+Agenda,+ChartIn,-ChartOut)  
  
process_agenda([],X,X).  
process_agenda([Edge|Agenda0],Chart0,Chart) :-  
    element(Edge,Chart0), !,  
    process_agenda(Agenda0,Chart0,Chart).  
process_agenda([Edge|Agenda0],Chart0,Chart) :-  
    Chart1=[Edge|Chart0],  
    %  
    predict(Edge, PAgenda),  
    append(PAgenda, Agenda0, Agenda1),  
    %  
    complete(Edge, Chart1, CAgenda),  
    append(CAgenda, Agenda1, NewAgenda),  
    process_agenda(NewAgenda, Chart1, Chart).
```

23

```
scan([],N,N,Chart,Chart).  
scan([W|Ws],JminOne,N,Chart0,Chart) :-  
    J is JminOne+1,  
    setof(chart(JminOne,J,state(Cat,[])),  
          lex(Cat,W),  
          Agenda),  
    process_agenda(Agenda,Chart0,Chart1),  
    scan(Ws,J,N,Chart1,Chart).
```

24

```
predict(chart(_,J,state(_,[_B|_])),Agenda) :-  
    setof(chart(J,J,state(B, Gamma)),  
          (B ---> Gamma),  
          Agenda), !.  
predict(_,[]). % is passive edge or no matching grammar rule  
  
complete(chart(K,J,state(B,[])),Chart,Agenda) :-  
    setof(chart(I,J,state(A,Beta)),  
          element(chart(I,K,state(A, [B|Beta])), Chart),  
          Agenda), !.  
complete(_,_,[]). % is active edge or no matching chart edge
```

25

```
% -----  
% element(?Element,+List)  
  
element(X,[X|_]).  
element(X,[_|L]) :-  
    element(X,L).  
  
% -----  
% append(+List,?List,-List) or append(-List,?List,+List)  
  
append([],L,L).  
append([H|T],L,[H|R]) :-  
    append(T,L,R).
```

26